```
SetDirectory["/Volumes/MicroSD/Dropbox/PostDoc_SD/mathematica/David/surf"];
fileSurfs = {"surf_1_P_T_G.dat", "surf_2_P_T_G.dat"};
cacSurf = Table[ReadList[fileSurfs[[i]], {Number, Number, Number}],
   {i, 1, Length[fileSurfs]}];

(*Plot data for two surfaces*)
ListPlot3D[cacSurf]
```




```
(*Guess a quadratic form*)
eqnForm = aaa + bbb * T + ccc * p + ddd * T ^ 2 + eee * p ^ 2 + fff * p * T;

(*Fit quadrativ form to data*)
fittedSurf1 =
 NonlinearModelFit[cacSurf[[1]], eqnForm, {aaa, bbb, ccc, ddd, eee, fff}, {p, T}]
fittedSurf2 = NonlinearModelFit[cacSurf[[2]],
  eqnForm, {aaa, bbb, ccc, ddd, eee, fff}, {p, T}]
```

FittedModel[$-941.488 + 0.0146882\,p - 0.0000964396\,p^2 - 0.0000253548\,T + 3.77171\times10^{-7}\,p\,T - 2.4741\times10^{-8}\,T^2$]

FittedModel[$-941.483 + 0.0137978\,p - 0.0000543645\,p^2 - 0.0000262425\,T + 3.44166\times10^{-7}\,p\,T - 2.46611\times10^{-8}\,T^2$]

```
(*Contour plot quadratic form*)
ContourPlot[fittedSurf1[p, T], {p, -5, 15}, {T, 100, 1000}]
ContourPlot[fittedSurf2[p, T], {p, -5, 15}, {T, 100, 1000}]
```






```
(*Plot in 3D*)
p1 = Plot3D[fittedSurf1[p, T], {p, -4, 12},
    {T, 100, 2000}, PlotStyle → {Red, Directive[Opacity[0.4]]}];
p2 = ListPlot3D[cacSurf[[1]], PlotStyle → {Green, Directive[Opacity[0.4]]}];
p3 = Plot3D[fittedSurf2[p, T], {p, -4, 12},
    {T, 100, 2000}, PlotStyle → {Orange, Directive[Opacity[0.4]]}];
p4 = ListPlot3D[cacSurf[[2]], PlotStyle → {Blue, Directive[Opacity[0.4]]}];
```

```
(*Show plots*)
Show[{p1, p2}]
Show[{p3, p4}]
```





```
Show[{p1, p3}]
```




```
(*Inspect equations*)
surf1 = eqnForm /. fittedSurf1[[1]][[2]]
surf2 = eqnForm /. fittedSurf2[[1]][[2]]
```

$$-941.488 + 0.0146882\, p - 0.0000964396\, p^2 - 0.0000253548\, T + 3.77171 \times 10^{-7}\, p\, T - 2.4741 \times 10^{-8}\, T^2$$

$$-941.483 + 0.0137978\, p - 0.0000543645\, p^2 - 0.0000262425\, T + 3.44166 \times 10^{-7}\, p\, T - 2.46611 \times 10^{-8}\, T^2$$

```
(*Solve intersections*)
surfDiff = surf1 - surf2
intersection = Solve[surfDiff == 0, {T, p}] // Simplify
```

$$-0.00477935 + 0.00089033\, p - 0.0000420751\, p^2 + 8.87745 \times 10^{-7}\, T + 3.30044 \times 10^{-8}\, p\, T - 7.98877 \times 10^{-11}\, T^2$$

... Solve: Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result.

... Solve: Equations may not give solutions for all "solve" variables.

$$\{\{p \to 10.5802 + 0.000392208\, T - 3.3646 \times 10^{-37} \sqrt{-1.4568 \times 10^{73} + 2.5969 \times 10^{71}\, T - 1.54133 \times 10^{67}\, T^2}\}, \{p \to 10.5802 + 0.000392208\, T + 3.3646 \times 10^{-37} \sqrt{-1.4568 \times 10^{73} + 2.5969 \times 10^{71}\, T - 1.54133 \times 10^{67}\, T^2}\}\}$$

```
intersection
```

$$\left\{\left\{p \to 10.5802 + 0.000392208\,T - 3.3646\times10^{-37}\sqrt{-1.4568\times10^{73} + 2.5969\times10^{71}\,T - 1.54133\times10^{67}\,T^2}\right\}, \left\{p \to 10.5802 + 0.000392208\,T + 3.3646\times10^{-37}\sqrt{-1.4568\times10^{73} + 2.5969\times10^{71}\,T - 1.54133\times10^{67}\,T^2}\right\}\right\}$$

```
(*Plot the intersection - note it won't be value for this whole range*)

Plot[{intersection[[1]][[1]][[2]], intersection[[2]][[1]][[2]]},
 {T, 10, 1500}, AxesLabel → {"Temperature (K)", "Pressure"}]
```




```
(*Find out the lowest temperature -
 ie solve for when is the top curve equal to the bottom curve*)

topCurve = intersection[[2]][[1]][[2]]
bottomCurve = intersection[[1]][[1]][[2]]
Solve[topCurve == bottomCurve, T][[1]]
```

$$10.5802 + 0.000392208\,T + 3.3646\times10^{-37}\sqrt{-1.4568\times10^{73} + 2.5969\times10^{71}\,T - 1.54133\times10^{67}\,T^2}$$

$$10.5802 + 0.000392208\,T - 3.3646\times10^{-37}\sqrt{-1.4568\times10^{73} + 2.5969\times10^{71}\,T - 1.54133\times10^{67}\,T^2}$$

$$\{T \to 56.2854\}$$

```
(*56 is pretty close to 59!*)
```